>

Triple Integrals

```
> restart:with(plots):
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,
  70]);
```

Finding the Volume of a Quarter Sphere.

We wish to find the volume of a quarter sphere of radius 1 by using a triple integral. We will place the quarter sphere above the xy-plane, extending from 0 to 1 on the x-axis and -1 to 1 on the y-axis We view the graph.

```
> plot3d([x,y,sqrt(1-x^2-y^2)],x=0..1,y=-1..1,style=patchnogrid,
  scaling=constrained,orientation=[30,20],numpoints=5000);
```



Next we write and evaluate the triple integral giving the volume.

```
> sphere_volume:=Int(Int(Int(1,y=-sqrt(1-x^2-z^2)..sqrt(1-x^2-z^2)
  ),x=0..sqrt(1-z^2)),z=0..1)=int(int(int(1,y=-sqrt(1-x^2-z^2)..
  sqrt(1-x^2-z^2)),x=0..sqrt(1-z^2)),z=0..1);
```

$$sphere\_volume := \int_0^1 \int_0^{\sqrt{1-z^2}} \int_{-\sqrt{1-x^2-z^2}}^{\sqrt{1-x^2-z^2}} 1 \, dy \, dx \, dz = \frac{1}{3}\pi$$

That was quick. Now let's do it step by step from the inside out.

```
> I1:=Int(1,y=-sqrt(1-x^2-z^2)..sqrt(1-x^2-z^2))=int(1,y=-sqrt(1-
  x^2-z^2)..sqrt(1-x^2-z^2));
```

$$I1 := \int_{-\sqrt{1-x^2-z^2}}^{\sqrt{1-x^2-z^2}} 1 \, dy = 2\sqrt{1-x^2-z^2}$$

```
> I2:=Int(lhs(I1),x=0..sqrt(1-z^2))=int(rhs(I1),x=0..sqrt(1-z^2));
```

$$I2 := \int_{0}^{\sqrt{1-z^2}} \int_{-\sqrt{1-x^2-z^2}}^{\sqrt{1-x^2-z^2}} 1 \, dy \, dx = \frac{1}{2}\left(1-z^2\right)\pi$$

```
> sphere_volume:=Int(lhs(I2),z=0..1)=int(rhs(I2),z=0..1);
```

$$sphere\_volume := \int_{0}^{1} \int_{0}^{\sqrt{1-z^2}} \int_{-\sqrt{1-x^2-z^2}}^{\sqrt{1-x^2-z^2}} 1 \, dy \, dx \, dz = \frac{1}{3}\pi$$

Volume under a Tetrahedron.

We wish to integrate 2*x+3*y over the tetrahedron bounded by $2x + 3y + z = 6$ and the coordinate planes.

```
> p1:=plot3d(6-2*x-3*y,x=0..4,y=0..3,view=-.1..8,color=red,style=
  patchnogrid,transparency=.5):
  p2:=implicitplot3d(x=0,x=0..4,y=0..3,view=-.1..8,color=blue,
  style=patchnogrid):
  p3:=implicitplot3d(y=0,x=0..4,y=0..3,view=-.1..8,color=green,
  style=patchnogrid):
  p4:=plot3d(0,x=0..4,y=0..3,view=-.1..8,color=gold,style=
  patchnogrid):
  display(p1,p2,p3,p4);
```



We plot the base.

```
> plot(-2/3*x+2,x=0..3);
```



We use the fact that the base is vertically simple.

```
> V:=Int(Int(Int(2*x+3*y,z=0..6-2*x-3*y),y=0..-2/3*x+2),x=0..3)
  =int(int(int(2*x+3*y,z=0..6-2*x-3*y),y=0..-2/3*x+2),x=0..3);
```

$$V := \int_0^3 \int_0^{-\frac{2}{3}x+2} \int_0^{6-2x-3y} (2x+3y)\, dz\, dy\, dx = 18$$

Volume of the Intersection of Two Cylinders.

We wish to find the volume of the intersection of the cylinders $x^2 + z^2 = 1$ and $y^2 + z^2 = 1$. We visualize the graph.

```
> g1:=implicitplot3d( {x^2+z^2=1},x=-1..1,y=-1..1,z=-1..1,style=
  patchnogrid,color=red):
  g2:=implicitplot3d( {y^2+z^2=1},x=-1..1,y=-1..1,z=-1..1,style=
  patchnogrid,color=green):
  display(g1,g2);
```



We find the volume using a triple integral.

```
> volume:=Int(Int(Int(1,y=-sqrt(1-z^2)..sqrt(1-z^2)),x=-sqrt(1-z^2)..sqrt(1-z^2)),z=-1..1)=int(int(int(1,y=-sqrt(1-z^2)..sqrt(1-z^2)),x=-sqrt(1-z^2)..sqrt(1-z^2)),z=-1..1);
```

$$volume := \int_{-1}^{1}\int_{-\sqrt{1-z^2}}^{\sqrt{1-z^2}}\int_{-\sqrt{1-z^2}}^{\sqrt{1-z^2}} 1 \, dy \, dx \, dz = \frac{16}{3}$$

```
>
>
>
>
>
```